


```
/*
а) Добавьте в таблицу dbo.StateProvince поле CountryRegionName типа nvarchar(50)
*/
ALTER TABLE dbo.StateProvince
    ADD CountryRegionName NVARCHAR(50);
GO
```

Рисунок 1.1 – Задание 1-а (Решение)



dbo.StateProvince

Columns

- StateProvinceID (int, null)
- StateProvinceCode (nchar(3), null)
- CountryRegionCode (nvarchar(3), null)
- Name (nvarchar(50), null)
- TerritoryID (int, null)
- ModifiedDate (datetime, null)
- CountryNum (int, null)
- CountryRegionName (nvarchar(50), null)

Рисунок 1.2 – Задание 1-а (Результат)

```

/*
b) Объявите табличную переменную с такой же структурой как dbo.StateProvince и
заполните ее данными из dbo.StateProvince.
Заполните поле CountryRegionName данными из Person.CountryRegion поля Name
*/
DECLARE @StateProvinceTableVar
TABLE
(
    StateProvinceID INT,
    StateProvinceCode NCHAR(3),
    CountryRegionCode NVARCHAR(3),
    Name NVARCHAR(50),
    TerritoryID INT,
    ModifiedDate DATETIME,
    CountryNum INT,
    CountryRegionName NVARCHAR(50)
);

INSERT INTO @StateProvinceTableVar
SELECT
    Province.StateProvinceID,
    Province.StateProvinceCode,
    Province.CountryRegionCode,
    Province.Name,
    Province.TerritoryID,
    Province.ModifiedDate,
    Province.CountryNum,
    Region.Name
FROM dbo.StateProvince AS Province
JOIN Person.CountryRegion AS Region
ON Province.CountryRegionCode = Region.CountryRegionCode;

SELECT *
FROM @StateProvinceTableVar;

```

Рисунок 1.3 – Задание 1-b (Решение)

	StateProvinc...	StateProvinceC...	CountryRegionC...	Name	Territor...	ModifiedDate	CountryN...	CountryRegionName
1	5	AS	AS	American Samoa	1	2008-03-11 10:17:21.587	NULL	American Samoa
2	84	FR	FR	France	7	2002-06-01 00:00:00.000	NULL	France
3	34	MH	MH	Marshall Islands	9	2008-03-11 10:17:21.587	NULL	Marshall Islands
4	16	FM	FM	Micronesia	9	2008-03-11 10:17:21.587	NULL	Micronesia
5	38	MP	MP	Northern Mariana Islands	9	2008-03-11 10:17:21.587	NULL	Northern Mariana Islands
6	62	PW	PW	Palau	9	2008-03-11 10:17:21.587	NULL	Palau

Рисунок 1.4 – Задание 1-b (Результат)

```

/*
    c) Обновите поле CountryRegionName в dbo.StateProvince данными из табличной переменной
*/
UPDATE dbo.StateProvince
SET CountryRegionName = Source.CountryRegionName
FROM @StateProvinceTableVar as Source
WHERE dbo.StateProvince.StateProvinceID = Source.StateProvinceID;
GO

SELECT *
FROM dbo.StateProvince;
GO

```

Рисунок 1.5 – Задание 1-с (Решение)

	StateProvinc...	StateProvinceC...	CountryRegionC...	Name	Territor...	ModifiedDate	CountryN...	CountryRegionName
1	5	AS	AS	American Samoa	1	2008-03-11 10:17:21.587	NULL	American Samoa
2	84	FR	FR	France	7	2002-06-01 00:00:00.000	NULL	France
3	34	MH	MH	Marshall Islands	9	2008-03-11 10:17:21.587	NULL	Marshall Islands
4	16	FM	FM	Micronesia	9	2008-03-11 10:17:21.587	NULL	Micronesia
5	38	MP	MP	Northern Mariana Islands	9	2008-03-11 10:17:21.587	NULL	Northern Mariana Islands
6	62	PW	PW	Palau	9	2008-03-11 10:17:21.587	NULL	Palau

Рисунок 1.6 – Задание 1-с (Результат)

```

/*
    d) Удалите штаты из dbo.StateProvince, которые отсутствуют в таблице Person.Address
*/
DELETE FROM dbo.StateProvince
WHERE StateProvinceID NOT IN
(
    SELECT StateProvinceID
    FROM Person.Address
);
GO

SELECT *
FROM dbo.StateProvince;
GO

```

Рисунок 1.7 – Задание 1-d (Решение)

StateProvinc...	StateProvinceC...	CountryRegionC...	Name	Territor...	ModifiedD...	CountryN...	CountryRegionNa...
-----------------	-------------------	-------------------	------	-------------	--------------	-------------	--------------------

Рисунок 1.8 – Задание – 1-d (Результат)

```

/*
e) Удалите поле CountryRegionName из таблицы, удалите все созданные ограничения и значения по умолчанию.
Имена ограничений вы можете найти в метаданных.
Имена значений по умолчанию найдите самостоятельно, приведите код, которым пользовались для поиска.
*/
SELECT sys.objects.name
FROM sys.tables
JOIN sys.objects
ON sys.tables.object_id = sys.objects.parent_object_id
WHERE sys.tables.name = 'StateProvince' AND sys.tables.schema_id = schema_id('dbo');

ALTER TABLE dbo.StateProvince
DROP COLUMN CountryRegionName;

ALTER TABLE dbo.StateProvince
DROP CONSTRAINT U_Name, CH_CountryRegionCode, D_ModifiedDate;

```

Рисунок 1.9 – Задание 1-е (Решение)




Рисунок 1.10 – Задание 1-е (Результат)

```

/*
f) Удалите таблицу dbo.StateProvince
*/
DROP TABLE dbo.StateProvince;

```

Рисунок 1.11 – Задание 1-f (Решение)




Рисунок 1.12 – Задание 1-f (Результат)

```

/*
a) Выполните код, созданный во втором задании второй лабораторной работы.
Добавьте в таблицу dbo.StateProvince поля SalesYTD MONEY и SumSales MONEY.
Также создайте в таблице вычисляемое поле SalesPercent,
вычисляющее процентное выражение значения в поле SumSales от значения в поле SalesYTD.
*/
ALTER TABLE dbo.StateProvince
ADD
    SalesYTD MONEY,
    SumSales MONEY,
    SalesPercent AS (SumSales / SalesYTD * 100);
GO

```

Рисунок 2.1 – Задание 2-а (Решение)




Рисунок 2.2 – Задание 2-а (Результат)

```
/*
b) Создайте временную таблицу #StateProvince, с первичным ключом по полю StateProvinceID.
    Временная таблица должна включать все поля таблицы dbo.StateProvince за исключением поля SalesPercent.
*/
CREATE TABLE #StateProvince
(
    StateProvinceID INT PRIMARY KEY,
    StateProvinceCode NCHAR(3),
    CountryRegionCode NVARCHAR(3),
    Name NVARCHAR(50),
    TerritoryID INT,
    ModifiedDate DATETIME,
    CountryNum INT,
    SalesYTD MONEY,
    SumSales MONEY
);
GO
```

Рисунок 2.3 – Задание 2-b (Решение)




Рисунок 2.4 – Задание 2-b (Результат)

```
/*
с) Заполните временную таблицу данными из dbo.StateProvince.
   Поле SalesYTD заполните значениями из таблицы Sales.SalesTerritory.
   Посчитайте сумму продаж (SalesYTD) для каждой территории (TerritoryID) в таблице Sales.SalesPerson и
   заполните этими значениями поле SumSales.
   Подсчет суммы продаж осуществите в Common Table Expression (CTE).
*/
WITH SalesSum (TerritoryID, SalesSum)
AS
(
    SELECT
        TerritoryID,
        SUM(SalesYTD) as SalesSum
    FROM Sales.SalesPerson
    GROUP BY TerritoryID
)
INSERT INTO #StateProvince
SELECT
    dbo.StateProvince.StateProvinceID,
    dbo.StateProvince.StateProvinceCode,
    dbo.StateProvince.CountryRegionCode,
    dbo.StateProvince.Name,
    dbo.StateProvince.TerritoryID,
    dbo.StateProvince.ModifiedDate,
    dbo.StateProvince.CountryNum,
    Sales.SalesTerritory.SalesYTD,
    SalesSum.SalesSum
FROM dbo.StateProvince
JOIN Sales.SalesTerritory
ON Sales.SalesTerritory.TerritoryID = dbo.StateProvince.TerritoryID
JOIN SalesSum
ON SalesSum.TerritoryID = dbo.StateProvince.TerritoryID;

SELECT *
FROM #StateProvince;
```

Рисунок 2.5 – Задание 2-с (Решение)

	StateProvinc...	StateProvinceC...	CountryRegionC...	Name	Territor...	ModifiedDate	CountryN...	SalesYTD	SumSales
1	5	AS	AS	American Samoa	1	2008-03-11 10:17:21.587	NULL	7887186,7882	4502152,2674
2	16	FM	FM	Micronesia	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242
3	34	MH	MH	Marshall Islands	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242
4	38	MP	MP	Northern Mariana Islands	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242
5	62	PW	PW	Palau	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242
6	84	FR	FR	France	7	2002-06-01 00:00:00.000	NULL	4772398,3078	3121616,3202

Рисунок 2.6 – Задание 2-с (Результат)

```

/*
d) Удалите из таблицы dbo.StateProvince одну строку (где StateProvinceID = 5)
*/
DELETE FROM dbo.StateProvince
WHERE StateProvinceID = 5;

SELECT *
FROM dbo.StateProvince;

```

Рисунок 2.7 – Задание 2-d (Решение)

	StateProvinc...	StateProvinceC...	CountryRegionC...	Name	Territor...	ModifiedDate	CountryN...	SalesY...	SumSal...	SalesPerc...
1	84	FR	FR	France	7	2002-06-01 00:00:00.000	NULL	NULL	NULL	NULL
2	34	MH	MH	Marshall Islands	9	2008-03-11 10:17:21.587	NULL	NULL	NULL	NULL
3	16	FM	FM	Micronesia	9	2008-03-11 10:17:21.587	NULL	NULL	NULL	NULL
4	38	MP	MP	Northern Mariana Islands	9	2008-03-11 10:17:21.587	NULL	NULL	NULL	NULL
5	62	PW	PW	Palau	9	2008-03-11 10:17:21.587	NULL	NULL	NULL	NULL

Рисунок 2.8 – Задание 2-d (Результат)



```

/*
e) Напишите Merge выражение, использующее dbo.StateProvince как target, а временную таблицу как source.
Для связи target и source используйте StateProvinceID.
Обновите поля SalesYTD и SumSales, если запись присутствует в source и target.
Если строка присутствует во временной таблице, но не существует в target,
добавьте строку в dbo.StateProvince.
Если в dbo.StateProvince присутствует такая строка, которой не существует во временной таблице,
удалите строку из dbo.StateProvince.
*/
MERGE dbo.StateProvince AS TargetTable
USING #StateProvince AS SourceTable
ON TargetTable.StateProvinceID = SourceTable.StateProvinceID
WHEN MATCHED THEN
    UPDATE
    SET
        TargetTable.SalesYTD = SourceTable.SalesYTD,
        TargetTable.SumSales = SourceTable.SumSales
WHEN NOT MATCHED THEN
    INSERT
    VALUES
        (
            SourceTable.StateProvinceID,
            SourceTable.StateProvinceCode,
            SourceTable.CountryRegionCode,
            SourceTable.Name,
            SourceTable.TerritoryID,
            SourceTable.ModifiedDate,
            SourceTable.CountryNum,
            SourceTable.SalesYTD,
            SourceTable.SumSales
        )
WHEN NOT MATCHED BY SOURCE THEN
    DELETE;

SELECT *
FROM dbo.StateProvince
ORDER BY StateProvinceID;

```

Рисунок 2.9 – Задание 2-е (Решение)

	StateProvinc...	StateProvinceC...	CountryRegionC...	Name	Territor...	ModifiedDate	CountryN...	SalesYTD	SumSales	SalesPerc...
1	5	AS	AS	American Samoa	1	2008-03-11 10:17:21.587	NULL	7887186,7882	4502152,2674	57,08
2	16	FM	FM	Micronesia	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242	23,78
3	34	MH	MH	Marshall Islands	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242	23,78
4	38	MP	MP	Northern Mariana Islands	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242	23,78
5	62	PW	PW	Paleu	9	2008-03-11 10:17:21.587	NULL	5977814,9154	1421810,9242	23,78
6	84	FR	FR	France	7	2002-06-01 00:00:00.000	NULL	4772398,3078	3121616,3202	65,40

Рисунок 2.10 – Задание 2-е (Результат)